
Preface

COPYING AND DISTRIBUTION POLICY
This document is part of a series of notes titled
"Maxima by Example" and is made available
via the author’s webpage http://www.csulb.edu/˜woollett/
to aid new users of the Maxima computer algebra system.

NON-PROFIT PRINTING AND DISTRIBUTION IS PERMITTED.

You may make copies of this document and distribute them
to others as long as you charge no more than the costs of printing.

These notes (with some modifications) will be published in book form
eventually via Lulu.com in an arrangement which will continue
to allow unlimited free download of the pdf files as well as the option
of ordering a low cost paperbound version of these notes.

Feedback from readers is the best way for this series of notes to become more helpful to new users of Maxima. All
comments and suggestions for improvements will be appreciated and carefully considered.

LOADING FILES
The defaults allow you to use the brief version load(fft) to load in the
Maxima file fft.lisp.
To load in your own file, such as qfft.mac (used in this chapter),
using the brief version load(qfft), you either need to place
qfft.mac in one of the folders Maxima searches by default, or
else put a line like:

file_search_maxima : append(["c:/work3/###.{mac,mc}"],file_search_maxima )$

in your personal startup file maxima-init.mac (see Ch. 1, Introduction
to Maxima for more information about this).

Otherwise you need to provide a complete path in double quotes,
as in load("c:/work3/qfft.mac"),

We always use the brief load version in our examples, which are generated
using the Xmaxima graphics interface on a Windows XP computer, and copied
into a fancy verbatim environment in a latex file which uses the fancyvrb
and color packages.

Maxima, a Computer Algebra System.
Version 5.19.0 (2009) using Lisp GNU Common Lisp (GCL) GCL 2.6.8
(aka GCL). http://maxima.sourceforge.net/

The homemade function fll(x) (first, last, length) is used to return the first and last elements of lists (as well as the length), and is
defined in the Ch.1 utility file mbe1util.mac. We will include a reference to this definition when working with lists.

This function is defined by:

fll(x) := [first(x),last(x),length(x)]$
declare(fll,evfun)$

The author would like to thank the Maxima developers for their friendly help via the Maxima mailing list.

11.1 Examples of the Use of the Fast Fourier Transform Functions fft and inverse fft

We discuss the use of Maxima’s fast Fourier transform package fft.lisp, which has been rewritten (effective with
version 5.19) to allow list input and output (rather than being restricted to array input and output).

We first present five simple examples of using Maxima’s fast Fourier transform functions fft and inverse fft. We then
discuss our notation and the utility functions available in the Ch.11 file qfft.mac. We then present a derivation of the
discrete Fourier transformation pairs, using Maxima’s conventions.

11.1.1 Example 1: FFT Spectrum of a Monochromatic Signal

We load in the Maxima package fft.lisp and also our own package qfft.mac.

(%i1) load(fft);
(%o1) C:/PROGRA˜1/MA89DF˜1.0/share/maxima/5.19.0/share/numeric/fft.lisp
(%i2) load(qfft);
(%o2) c:/work3/qfft.mac

Our first example uses the simplest possible signal which still contains some information, a signal having one intrinsic
frequency. We assume the signal is F(t) = cos(6 π t). This signal has an angular frequency ω = 6 π = 2 π f where f
is the frequency in Hertz (ie., in sec−1). Thus the frequency f = 3 sec−1. We bind the symbol e to this signal expression
for later use:

(%i3) e : cos ( 6*%pi*t )$

Let ns be the (even) number of function samples, and fs be the sampling frequency.

The qfft package function nyquist(ns, fs) computes the time interval between signal samples dt = 1/fs, the Nyquist
integer knyq = ns/2, the frequency resolution df = fs/ns, and the Nyquist frequency
fnyq = knyq * df = fs/2.

The product of the time interval dt between samples of the signal and the frequency resolution df is always the inverse
of the total number of signal samples ns:

dt * df = 1 / ns

We select fs and ns to satisfy two conditions. The only intrinsic frequency of the signal is f0 = 3 s−1. First we need
fs > 2 f0 which means that fs > 6 s−1. Secondly we need the frequency resolution df to satisfy df < f0, which will be
satisfied if we choose df such that f0 = 3df , or df = 1 hertz. But then df = fs / ns = 1, or fs = ns, so
the first condition becomes ns > 6, but we also need ns to be an even number of the form 2m for some integer m, so we
choose ns = 8.

We then bind both ns and fs to the value 8 and assign dt (the sampling interval) to the first element of the list produced
by the package function nyquist( ns, fs ), which also prints out dt, knyq, fnyq, and df.

(%i4) ( ns : 8, fs : 8 )$
(%i5) dt : first (nyquist (ns,fs));
sampling interval dt = 0.125
Nyquist integer knyq = 4
Nyquist freq fnyq = 4.0
freq resolution df = 1.0
(%o5) 0.125

We then use the package function sample(expr, var, ns, dvar) to generate a list of floating point numbers as
the expression expr is evaluated ns times, generating the list (with dvar being the time step dt here)
[F(0),F(dt),F(2*dt),...,F((ns-1)*dt )] holding the values F(m*dt), for m = 0,1,2,...,ns-1. More
details of the syntax of the qfft.mac package functions can be found in Section 11.3. Recall that we have just bound
ns and dt to numerical values and that the expression e depends on the variable t.

(%i6) flist : sample (e,t,ns,dt);
(%o6) [1.0, - 0.707, - 1.83691E-16, 0.707, - 1.0, 0.707, 5.51073E-16, - 0.707]

We see that elements three, F(2*dt), and seven, F(6*dt), are tiny numbers of the order of the default floating
point errors, and are numerically equivalent to zero.

We first make a plot of both the given function (the signal) and a list of points [m*dt, F(m*dt)] constructed using
the package function vf (flist, dt). Recall that the signal is first sampled at t = 0.

(%i7) tflist : vf (flist,dt);
(%o7) [[0, 1.0], [0.125, - 0.707], [0.25, - 1.83691E-16], [0.375, 0.707],

[0.5, - 1.0], [0.625, 0.707], [0.75, 5.51073E-16], [0.875, - 0.707]]

We let tmax be the “sampling time”, the number of signal samples times the time interval between samples.

(%i8) tmax : ns*dt;
(%o8) 1.0
(%i9) plot2d([e ,[discrete,tflist]], [t,0,tmax],

[style,[lines,3],[points,3,2,1]],
[legend,false])$

Figure 1: F(t) = cos(6 π t) with Sample Points

Now that we have looked at both the signal and sample signal values (on the same plot), we next look at the fast Fourier
frequency spectrum implied by this single frequency signal sample. We first define glist to be the fast fourier trans-
form of flist

(%i10) glist : fft (flist);
(%o10) [- 2.34662E-17, 1.11022E-16 - 3.30546E-17 %i,
1.52656E-16 %i - 4.59227E-17, 0.5 - 1.18171E-16 %i, 1.15312E-16,
2.16746E-16 %i + 0.5, - 1.52656E-16 %i - 4.59227E-17,
5.55112E-17 - 6.55197E-17 %i]
(%i11) fchop (%);
(%o11) [0.0, 0.0, 0.0, 0.5, 0.0, 0.5, 0.0, 0.0]

in which we have used Maxima’s fast fourier transform function fft. We then used our package function fchop to set
tiny floating point numbers to zero. The qfft package function current_small() prints out the current setting of the
qfft.mac parameter _small% used by fchop.

(%i12) current_small()$
current default small chop value = 1.0E-13

The first element of glist corresponds to k = 0, and the list glist contains the values of the fast Fourier transform amplitudes
G(k*df) for k = 0, 1, 2, ..., ns -1, where df is the frequency resolution (for this example, df = 1 hertz). We
call the value G(k*df) the “fast Fourier transform amplitude” corresponding to the frequency f = k*df.

If we look at the chopped version, we see that the first non-zero fast fourier transform amplitude occurs at element four, which
corresponds to k = 3, which corresponds to the frequency f = k*df = 3 * 1 = 3 hertz. In a later section exploring the
basic ideas of the fast Fourier transform, we explain why all the usable spectrum information is contained in the interval k = 0 to
k = knyq = 4.

To make a simple point plot of the fourier amplitudes, we first use the package function kg to make a list of the points [k, G(k*df)]
out to the Nyquist integer knyq. Since we want real numbers for a plot, this function takes the absolute value of each Fourier ampli-
tude and also chops tiny numbers.

(%i13) kglist : kg (glist);
(%o13) [[0, 0.0], [1, 0.0], [2, 0.0], [3, 0.5], [4, 0.0]]
(%i14) plot2d ( [discrete, kglist],[x,0,5],[y,0,0.8],

[style,[points,5,1,1]],[xlabel,"k"],
[ylabel," "],[gnuplot_preamble,"set grid;"])$

which produces the simple plot

Figure 2: Spectrum of cos(6 π t)

We can also use plot2d to make a simple histogram, using kglist.

(%i15) vbars : makelist ( [discrete,
[[kglist[i][1],0],[kglist[i][1],kglist[i][2]]]] ,

i,1,length(kglist) );
(%o15) [[discrete, [[0, 0], [0, 0.0]]], [discrete, [[1, 0], [1, 0.0]]],
[discrete, [[2, 0], [2, 0.0]]], [discrete, [[3, 0], [3, 0.5]]],
[discrete, [[4, 0], [4, 0.0]]]]
(%i16) plot2d ( vbars,[y,0,0.8],[style,[lines,5,1]],

[ylabel," "],[xlabel," k "],[legend,false],
[gnuplot_preamble,"set grid;"] )$

which produces the plot

Figure 3: Line Spectrum of cos(6 π t)

The production of such a frequency space spectrum plot can be automated by accepting the fast Fourier transform list glist
and defining a Maxima function spectrum (glist, nlw, ymax) which would be used with the specific syntax spectrum
(glist, 5, 0.8) to obtain a histogram similar to the above but cleaner. (nlw is the line width and ymax is the vertical extent
of the canvas starting at y = 0.)

We have designed spectrum to also allow the expanded syntax spectrum (glist, nlw, ymax, k1, k2) which zooms in on
the interval k1 <= k <= k2, where k2 <= knyq. In addition, we have avoided, in spectrum, creating vertical bars
when a fast Fourier transform amplitude is less than a very small number.

The entry

(%i17) spectrum (glist, 5, 0.8 )$

produces the plot

Figure 4: spectrum ( glist, 5, 0.8 )

We will use spectrum, included in qfft.mac, in our following examples as a quick way to look at the frequency spectrum.

Finally, let’s use the Maxima function inverse fft to apply the inverse fast Fourier transform to the list of fast Fourier
transform amplitudes glist, and see how closely the result matches the original signal sample list flist.

(%i18) flist1 : inverse_fft(glist);
(%o18) [1.0, 1.11022E-16 %i - 0.707, 2.24949E-32 %i - 1.83691E-16,
6.77259E-17 %i + 0.707, - 1.0, 0.707 - 1.11022E-16 %i,
5.51073E-16 - 2.24949E-32 %i, - 6.77259E-17 %i - 0.707]
(%i19) fchop(%);
(%o19) [1.0, - 0.707, 0.0, 0.707, - 1.0, 0.707, 0.0, - 0.707]
(%i20) fchop( flist);
(%o20) [1.0, - 0.707, 0.0, 0.707, - 1.0, 0.707, 0.0, - 0.707]
(%i21) lmax ( abs (flist1 - flist));
(%o21) 1.30049E-16

We see that inverse fft (glist) recovers our original signal sample list flist to within floating point errors.

11.1.2 Example 2: FFT Spectrum of a Sum of Two Monochromatic Signals

Almost all signals will contain more than one intrinsic frequency, and to recover a portion of the frequency spectrum with
fidelity, we need to satisfy the two conditions

fs > 2 fhigh and df ≤ flow, (11.1)

in which flow is the lowest intrinsic frequency to be identified, and fhigh is the highest intrinsic frequency to be identified.
(Again, fs is the sampling frequency, and df is the frequency resolution).

We assume now that the signal is F(t) = cos(2 π t) + sin(4 π t). We thus have a signal with the frequencies f1 = 1 s−1

and f2 = 2 s−1. With fs being the sampling frequency, and ns being the number of signal samples, we require fs > 2 fhigh = 4, as
well as df ≤ flow = 1. If we choose flow = 3df , then df = 1/3 = fs/ns, or fs = ns/3. So the first condtion then implies
we need ns/3 > 4, or ns > 12. Since ns also should be equal to 2m for some integer m, we choose ns = 16. Then fs = 16/3
and df = fs/ns = 1/3 hertz.

(%i1) ( load(fft), load(qfft) )$
(%i2) e : cos(2*%pi*t) + sin(4*%pi*t)$
(%i3) (ns : 16, fs : 16/3)$
(%i4) dt : first (nyquist (ns,fs));
sampling interval dt = 0.188
Nyquist integer knyq = 8
Nyquist freq fnyq = 2.6667
freq resolution df = 0.333
(%o4) 0.188
(%i5) flist : sample (e,t,ns,dt)$
(%i6) %,fll;
(%o6) [1.0, - 0.324, 16]
(%i7) tmax: ns*dt;
(%o7) 3.0
(%i8) tflist : vf (flist,dt)$
(%i9) %,fll;
(%o9) [[0, 1.0], [2.8125, - 0.324], 16]
(%i10) plot2d([e ,[discrete,tflist]], [t,0,tmax],

[style,[lines,3],[points,3,2,1]],
[legend,false])$

We have used our utility function fll described in the preface, which returns the first and last element of a list, as well as the length
of the list. The plot thus produced is

Figure 5: cos(2 π t) + sin(4 π t) with Sample Points

We next generate the list glist of fast Fourier transform amplitudes G(k*df) and use spectrum to look at the implied
frequency spectrum.

(%i11) glist : fft (flist)$
(%i12) %,fll;
(%o12) [- 7.94822E-17, 1.52656E-16 - 5.7151E-17 %i, 16]
(%i13) spectrum (glist,5,0.6)$

which produces the spectrum histogram:

Figure 6: Spectrum of cos(2 π t) + sin(4 π t)

Since df = 1/3, the line at k = 3 corresponds to the frequency f = k*df = 3*(1/3) = 1 Hertz and the line at
k = 6 corresponds to the frequency f = k*df = 6*(1/3) = 2 Hertz.



10

11.1.3 Example 3: FFT Spectrum of a Rectangular Wave

A rectangular wave as a function of time t with period equal to 64 sec which cycles between plus and minus 1 can be
constructed from floor and mod in the form: rwave(t) := 2*mod(floor(t/32),2) - 1.

For t = 0, t/32 = 0, floor(t/32) = floor(0) = 0, mod(0,2) = 0, so rwave(0) = -1.

(%i1) rwave(t):= 2*mod(floor(t/32),2) -1 $
(%i2) [floor(0),mod(0,2),rwave(0)];
(%o2) [0, 0, - 1]

For t = 32, floor(32/32) = floor(1) = 1, mod(1,2) = 1, rwave(32) = 1.

(%i3) [floor(1),mod(1,2),rwave(32)];
(%o3) [1, 1, 1]

For t = 64, floor(64/32) = floor(2) = 2, mod(2,2) = 0, rwave(64) = -1.

(%i4) [floor(2),mod(2,2),rwave(64)];
(%o4) [2, 0, - 1]

Hence rwave(0 + 64) = rwave(0) and rwave(t) has a period equal to 64 sec.

We first look at the rectangular wave signal with a plot.

(%i5) plot2d(rwave(t),[t,0,128],[y,-1.5,1.5],
[ylabel," "],[style,[lines,5]],
[gnuplot_preamble,"set grid;set zeroaxis lw 2;"])$

Figure 7: Rectangular Wave with Period 64 sec.

If we consider sampling this signal at intervals dt = 1 sec, the sampling frequency will be fs = 1/dt =
1 hertz = 1 per sec = 64 per cycle. The lowest intrinsic frequency of this signal is f_low =
1/64 hertz—, corresponding to the period of the rectangular wave. If we choose 4*df = f_low, then
df = 1/256 = fs/ns = 1/ns, so ns = 256. Due to the sharp corners of a rectangular wave we expect many high
frequency components to be present in the spectrum.

We thus try the combination, ns = 256, and fs = 1 hertz.

(%i6) (ns:256, fs:1)$
(%i7) (load(fft),load(qfft) )$
(%i8) dt:first(nyquist(ns,fs));
sampling interval dt = 1.0
Nyquist integer knyq = 128
Nyquist freq fnyq = 0.5
freq resolution df = 0.00391
(%o8) 1.0
(%i9) flist : sample(rwave(t),t,ns,dt)$
(%i10) fll (flist);
(%o10) [- 1.0, 1.0, 256]
(%i11) makelist (flist[i],i,1,10);
(%o11) [- 1.0, - 1.0, - 1.0, - 1.0, - 1.0, - 1.0, - 1.0, - 1.0, - 1.0, - 1.0]
(%i12) tmax: ns*dt;
(%o12) 256.0
(%i13) tflist : vf (flist,dt)$
(%i14) fll (tflist);
(%o14) [[0, - 1.0], [255.0, 1.0], 256]
(%i15) plot2d([rwave(t) ,[discrete,tflist]], [t,0,tmax],

[y,-1.5,1.5],[ylabel," "],
[style,[lines,3],[points,1,0,1]],
[legend,false])$

The sample_plot invocation produces the black points of the sample on top and bottom of the rectangular wave shown
here:

Figure 8: Black Sample Points with Rectangular Wave

Next we look at the signal frequency spectrum from k = 0 to k = knyq = 128. Since f = k*df, and df = 1/256
hertz, the low fundamental intrinsic frequency f_0 = 1/64 hertzwill be located at k = 4, and the maximum possible
frequency component will be fnyq = knyq*df = 0.5 hertz = 32*f_0. The spectrum plot is generated by passing
the list flist containing ns = 256 signal samples, along with nlw = 3, and ymax = 0.7 to spectrum:

(%i16) glist : fft (flist)$
(%i17) %,fll;
(%o17) [0.0, 0.0, 256]
(%i18) spectrum (glist,3,0.7)$

which produces the spectrum plot:

Figure 9: Spectrum from k = 0 to knyq = 128

The plot shows lines at k = 4, 12, 20, 28, 36, ..... Since f = k*df and f_0 = 4*df is the fundamental
frequency, the frequencies present are f_0, 3*f_0, 5*f_0, 7*f_0,... which is the “fundamental” plus odd har-
monics of the fundamental. We know there must be high frequency components to account for the sharp corners of the
signal.

11.1.4 Example 4: FFT Spectrum Sidebands of a Tone Burst Before and After Filtering

A tone burst signal consisting of a sine wave begins at t = 0 and ends abruptly after forty five seconds. The fast Fourier
frequency spectrum of such a signal, when sampled over the duration of the sine wave plus some time following the sine
wave end, will contain “sideband” frequencies above and below the intrinsic frequency of the sine wave. Many signal
filters can be used to allow the easy identification of the intrinsic sine wave frequency by surpressing the sideband frequen-
cies. We use for the tone burst the sine wave: sin(2 π t/5) during the time interval 0 ≤ t ≤ 45 sec, and then abruptly
dropping to 0 for t > 45 sec. The intrinsic frequency of the sine wave is f0 = 1/5 hertz, and the corresponding
period is 5 sec, thus the sine wave lasts for nine periods. We will use the following idealized model of such a signal by
ignoring the actual time needed to end the sine wave.

(%i1) sig(t) := if t < 45 then sin(2*%pi*t/5.0) else 0$

We will use this definition only for t ≥ 0 in our work. With fs the sampling frequency, if we want about 10 samples/cycle (ie.,
10 per 5 sec), then we want period*fs = 10, or fs roughly equal to 2 hertz. We want the intrinsic, frequency f0 (
0.2 hertz) to not be too close to the left end of the frequency spectrum plot so we can see the low frequency sidebands as well

as the higher frequency sidebands.

Suppose we try requiring that f0 = 0.2 hertz = 50*df = 50*fs/ns. Solving for the number of signal samples ns, we
get ns = 50 * 2 * 5 = 500. To get a power of 2 we choose the closest such quantity, ns = 512. We then return to our f0
location requirement to solve for the resulting value of fs to get fs = 512/250 = 2.048 hertz.

(%i2) load (qfft)$
(%i3) (ns:512,fs:2.048)$
(%i4) dt : first(nyquist(ns,fs));
sampling interval dt = 0.488
Nyquist integer knyq = 256
Nyquist freq fnyq = 1.024
freq resolution df = 0.004
(%o4) 0.488
(%i5) tmax : ns*dt;
(%o5) 250.0

We can plot a Maxima function defined with the if then else construct without problem since plot2d evaluates its arguments.

(%i6) plot2d ( sig(t),[t,0,tmax],[y,-1.5,1.5],
[style,[lines,1,0]],[ylabel," "],[nticks,100],
[legend,false],[gnuplot_preamble,"set grid;"])$

which produces the plot:

Figure 10: Sine Wave Tone Burst, Period = 5 sec, Duration = Nine Periods

Using the qfft package function sample with a Maxima function defined with the if...then construct produces a list
each element of which includes if then else, which can be used with plot2d. However, we will include an extra
evaluation in defining our flist so we can see the numerical values as a list. Note that the syntax expr, fll; causes an
extra evaluation, so to be careful we need to use the syntax fll (expr).

(%i7) flist : sample (sig(t),t,ns,dt)$
(%i8) fll (flist);
(%o8) [if 0.0 < 45.0 then 0.0 else 0.0,

if 249.51 < 45.0 then - 0.576 else 0.0, 512]
(%i9) flist : ev (flist)$
(%i10) fll (flist);
(%o10) [0.0, 0.0, 512]
(%i11) makelist(flist[i],i,1,10);
(%o11) [0.0, 0.576, 0.942, 0.964, 0.634, 0.0736, - 0.514, - 0.914, - 0.981,

- 0.69]
(%i12) tflist : vf (flist,dt)$
(%i13) fll (tflist);
(%o13) [[0, 0.0], [249.51, 0.0], 512]

We now use plot2d to show our sample points on top of our tone burst signal (both in black).

(%i14) plot2d([sig(t) ,[discrete,tflist]], [t,0,tmax],
[y,-1.5,1.5],[ylabel," "],

[style,[lines,1,0],[points,1,0,1]],
[legend,false])$

which produces the plot:

Figure 11: Signal and Sample Points Which Include 0’s for t > 45 sec

We now plot the unfiltered tone burst spectrum after generating glist with fft.

(%i15) load (fft)$
(%i16) glist : fft (flist)$
(%i17) fll (glist);
(%o17) [- 8.08763E-5, 0.00174 - 0.00286 %i, 512]
(%i18) spectrum (glist, 2, 0.1 )$

which produces the spectrum plot: We see strong sideband frequencies on both the low and high side of the intrinsic

Figure 12: Frequency Spectrum of Unfiltered Tone Burst

frequency f0 = 50*df, which is the highest peak at k = 50 (f = k*df = 50*(1/250) = 1/5 = 0.02). The sidebands
are a mathematical artifact of the finite duration of the sine wave tone burst with its abrupt beginning and end.

A windowing filter which smooths out the beginning and end of the “burst envelope” is used by defining a smoothed
envelope signal as the product of the unsmoothed signal and a suitable windowing function. The von Hann window
employs a sin2 pinch envelope.

(%i19) hannw(x,m) := sin(%pi*(x-1)/(m-1))ˆ2$
(%i20) sig_w(t) := hannw(t,45)*sig(t)$
(%i21) plot2d ( sig_w(t),[t,0,tmax],[y,-1.5,1.5],

[style,[lines,1,0]],[ylabel," "],[nticks,100],
[legend,false],[gnuplot_preamble,"set grid;"])$

which shows the filtered tone burst

Figure 13: Filtered Tone Burst

Using now the filtered tone burst signal sig_w(t), we construct a filtered signal sample list flist_w and look at the altered fast
Fourier spectrum (remember dt has been bound to 0.488 and ns to 512):

(%i22) flist_w : sample (sig_w(t),t,ns,dt)$
(%i23) fll (flist_w);
(%o23) [0.00509 (if 0.0 < 45.0 then 0.0 else 0.0),

0.799 (if 249.51 < 45.0 then - 0.576 else 0.0), 512]
(%i24) flist_w : ev (flist_w)$
(%i25) fll (flist_w);
(%o25) [0.0, 0.0, 512]
(%i26) makelist (flist_w[i],i,1,5);
(%o26) [0.0, 7.68319E-4, 2.63668E-6, 0.00106, 0.00293]
(%i27) glist_w : fft (flist_w)$
(%i28) makelist (glist_w[i],i,1,5);
(%o28) [1.59068E-5, 3.93191E-6 - 1.99372E-5 %i, - 1.83151E-5 %i - 1.84837E-5,

2.09989E-6 %i - 2.59175E-5, 1.69797E-5 %i - 9.67992E-6]
(%i29) spectrum (glist_w, 2, 0.1)$

and we see deletion of most of the sideband frequency peaks:

 0

 0.02

 0.04

 0.06

 0.08

 0.1

 0  50  100  150  200  250

 

 k 

Figure 14: Frequency Spectrum of the Filtered Tone Burst

11.1.5 Example 5: Cleaning a Noisy Signal using FFT Methods

We use the same signal as used in Example 2, Sec. (11.1.2), but add some random noise to it. Without the noise, the
signal is F(t) = cos(2 π t) + sin(4 π t). Thus the clean signal contains the frequencies f1 = 1 s−1 and f2 = 2 s−1,
and two corresponding periods τ 1 = 1/f1 = 1 sec and τ 2 = 1/f2 = 0.5 sec. To get a noisy looking signal we have to
sample and add noise a lot of times within one or two periods of the clean signal, where we use the maximum of the
intrinsic periods. Let’s try N = 512 samples over a time tmax = 2 sec. Then N δt = 2 sec, so fs = 1/dt = N/2 = 256.
The first condition on the sampling frequency is that fs > 2 fhigh, or fs > 4 s−1, which is certainly satisfied. The second
condition on fs is δf < flow, or df < 1 s−1, or fs/ns < 1 s−1, or fs < ns, which is also true. Hence we try ns = 512,
fs = 256.

(%i1) e : cos(2*%pi*t) + sin(4*%pi*t)$
(%i2) (load(fft), load(qfft))$
(%i3) [ns:512,fs:256]$
(%i4) dt : first(nyquist(ns,fs));
sampling interval dt = 0.00391
Nyquist integer knyq = 256
Nyquist freq fnyq = 128.0
freq resolution df = 0.5
(%o4) 0.00391
(%i5) flist : sample(e,t,ns,dt)$
(%i6) %,fll;
(%o6) [1.0, 0.951, 512]
(%i7) flist_noise : makelist(flist[j]+0.3*(-1.0+random(2.0)),j,1,ns)$
(%i8) %,fll;
(%o8) [1.2483, 1.0073, 512]
(%i9) tflist_noise : vf (flist_noise,dt)$
(%i10) %,fll;
(%o10) [[0, 1.2483], [1.9961, 1.0073], 512]
(%i11) plot2d ([discrete,tflist_noise],[y,-2,2],

[style,[lines,1]],[ylabel," "])$

which produces the noisy signal plot

Figure 15: Noisy Signal

One way to “clean” this signal is to set small numbers in the fast Fourier transform to zero and inverse transform back to
the time domain using inverse fft.

In order to “chop” small numbers in the fast Fourier transform amplitude list, we need to produce that list, which we call
glist noise from the noisy signal sample flist noise by using fft.

(%i12) glist_noise : fft (flist_noise)$
(%i13) %,fll;
(%o13) [- 0.0022, 5.09918E-4 %i - 0.00338, 512]

Before “chopping” small numbers in glist noise, we take a look at the fast Fourier frequency spectrum implied by
glist noise.

(%i14) spectrum (glist_noise,2,0.6)$

which produces the plot

Figure 16: Noisy Signal Frequency Spectrum

The dominant lines are still those corresponding to the two intrinsic frequencies of the clean signal we started with, but
there are many more frequencies present due to the noise.

To see more clearly the dominant line region, we show another plot with the integer k in the range (0, 10):

(%i15) spectrum (glist_noise,4,0.6,0,10)$

which produces the plot:

Figure 17: Noisy Signal Frequency Spectrum with k in (0,10)

We now use fchop1 with the value 0.2 to set small floating point numbers less than 0.2 to zero in the frequency space list
glist noise, and again use spectrum to look at the frequency spectrum associated with the chopped frequency space list.

(%i16) glist_noise_chop : fchop1(glist_noise,0.2)$
(%i17) %,fll;
(%o17) [0.0, 0.0, 512]
(%i18) spectrum (glist_noise_chop,2,0.6)$

which produces the plot

Figure 18: Chopped Frequency Spectrum

Here is the cleaned spectrum in the range k = (0,10):

Figure 19: Chopped Frequency Spectrum for k in (0,10)

We now create a cleaned up signal list using inverse fft on the chopped glist, transforming back to the time domain.

(%i19) flist_clean : inverse_fft ( glist_noise_chop )$
(%i20) %,fll;
(%o20) [2.22045E-16 %i + 1.0016, 0.952 - 3.16992E-15 %i, 512]
(%i21) flist_clean : realpart(flist_clean)$
(%i22) %,fll;
(%o22) [1.0016, 0.952, 512]

Since the inverse Fourier transform will often include small imaginary parts due to floating point error, we took care
to take the real part of the returned list before looking at the cleaned up signal. We now construct the list of points [t,
F clean]:

(%i23) tflist_clean : vf ( flist_clean, dt )$
(%i24) %,fll;
(%o24) [[0, 1.0016], [1.9961, 0.952], 512]

and plot first just the cleaned up signal points

(%i25) plot2d ([discrete,tflist_clean],[y,-2,2],
[style,[lines,1]],[ylabel," "])$

which produces

Figure 20: Cleaned Signal Points

We now show both the cleaned list points together with the original clean two frequency signal we started with, to show that the
cleaned up points lie right on top of the original two frequency signal.

(%i26) plot2d([e ,[discrete,tflist_clean]], [t,0,2],
[style,[lines,1],[points,1,0,1]],
[legend,false])$

which produces the plot

Figure 21: Cleaned Signal Points on Top of Original Clean Signal

We see that the inverse fast Fourier transform of the chopped glist frequency spectrum yields a cleaned up signal, as
desired.

11.2 Our Notation for the Discrete Fourier Transform and its Inverse

Given a real valued signal F(t) which is to be sampled N times at the moments (separated by equal time intervals δt)
t = 0, δt, 2 δt, ..., (N− 1) δt, one needs to select the two parameters N = 2m and fs, where the latter is called the
sampling frequency. The sampling time interval is then given by

δt =
1
fs

(11.2)

(or else use the data determined value of δt to calculate fs). Note that the fast fourier transform algorithm used by Maxima
assumes that the number of signal samples N is some integral power of 2, N = 4,8,16,32,64, ... so an experimental
sample might have to be padded with zeroes to achieve this condition on N. The sampling frequency fs should be greater
than twice the highest frequency component to be identified in the signal and the frequency resolution δf should be
smaller than the lowest frequency to be identified in the signal. Given the sampling frequency fs and the number of signal
samples N, the frequency resolution δf is given by

δf =
fs
N

(11.3)

We will motivate this definition below. Assuming this definition, we then require that

fs
N

< flow (11.4)

The sampling frequency fs thus needs to satisfy the two conditions:

2 fhigh < fs < Nflow (11.5)

A convenient choice which automatically satisfies the low frequency part of these conditions is to arrange that

flow = n δf , (11.6)

where n = 3 or 4, say. Then that choice determines the frequency resolution to be used δf = flow/n, and from the
definition of δf , Eq.(11.3), this requires that

fs = Nflow/n (11.7)

and then Eq.(11.5) implies the condition on N:

N >
2n fhigh

flow
(11.8)

In the simple case that flow = fhigh = f0 Eq.(11.7) becomes

fs = N
f0
n

, (11.9)

Eq.(11.5) becomes
N > 2n, (11.10)

and
δf =

f0
n

(11.11)

In Example 1, the signal frequency is f0 = 3, and we chose n = 3. Then, Eq.(11.9) implies that fs = N, Eq.(11.10)
implies that N > 6, and Eq.(11.11) implies that δf = 1. Since we need N = 2m as well, we chose fs = N = 8.

In Example 2, flow = 1 and fhigh = 2, and we again chose n = 3. Then, Eq.(11.6) implies that δf = 1/3, Eq.(11.7)
implies that fs = N/3, and Eq.(11.8) implies that N > 12. Since we need N = 2m as well, we chose N = 16 and this
forces fs = 16/3.

The N real numbers F(0), F(δt), F(2 δt), ...,F(m δt), ...,F((N− 1) δt) can be used to define N complex numbers
G(k δf), where k = 0, 1, ..., (N− 1), according to (this incorporates Maxima’s conventions):

G(k δf) =
1
N

N−1∑

m=0

F(m δt) e−2π imk/N (11.12)

The Maxima conventions include where to put the factor of 1/N and what sign to use in the exponent argument. Our
notation uses i to stand for the pure imaginary number

√−1. Common engineering notation uses j for
√−1.

Equation (11.12) can be exactly inverted to arrive at an expression for the values of the original signal at N discrete times
in terms of the N values of the discrete Fourier transform.

F(m δt) =
N−1∑

k=0

G(k δf) e2π i km/N (11.13)

where m = 0, 1, ..., (N− 1).

Using Eqs. (11.3) and (11.2), we make the replacement

1
N

= δf δt (11.14)

in Equations (11.12) and (11.13) to get

G(fk) =
1
N

N−1∑

m=0

F(tm) e−2π i fk tm (11.15)

and

F(tm) =
N−1∑

k=0

G(fk) e2π i tm fk (11.16)

where fk = k δf and tm = m δt.

A simpler looking set of transform pairs can be achieved by letting Fm = F(tm) and Gk = G(fk), in terms of which
Equations (11.12) and (11.13) become

Gk =
1
N

N−1∑

m=0

Fm e−2π imk/N (11.17)

and

Fm =
N−1∑

k=0

Gk e2π i km/N (11.18)

We can use
e−2π imk = (e−2π i k)m = (−1)m = 1 (11.19)

to show that the fast Fourier amplitudes have the periodicity N

Gk+N = Gk. (11.20)

We can now formally admit negative frequencies by letting k take on negative integral values, and setting k = −N/2 we
then get

GN/2 = G−N/2. (11.21)

This means that the amplitude corresponding to the “Nyquist frequency”

fNyquist =
N
2

δf (11.22)

is the same complex number as the amplitude corresponding to the frequency −fNyquist.
In the qfft package, the function nyquist calculates what we call the “Nyquist integer” kNyquist, which is just

kNyquist =
N
2

, (11.23)

in terms of which
fNyquist = kNyquist δf (11.24)

Likewise we can show that
GN/2+1 = G−N/2+1. (11.25)

which means that the amplitude corresponding to the frequency (N/2 + 1) δf = fNyquist + δf is the same complex
number as the amplitude corresponding to the frequency −fNyquist + δf . The only useful part of the spectrum in that
contained in the frequency interval between zero and fNyquist − δf , ie., in the range
k = 0,1,2, ..., (N/2)− 1.

In a similar manner we can show that Fm+N = Fm, or that F(tm + T) = F(tm), where

T = N δt =
N
fs

=
1
δf

, (11.26)

so that the fast Fourier amplitudes describe a signal which has the basic inevitable long period T no matter what other
shorter periods (and correspondingly higher frequencies) are also present in the signal. This low frequency, long period
property is an artifact of the approximate nature of Equations (11.12) and (11.13).

The fast fourier transform and its inverse should be considered as a distinct type of transform pair rather than as an
approximation to either a Fourier series expansion or a Fourier integral expression of a continuous spectrum. The basic
idea of the fast Fourier transform is that one has waited long enough for a physical system to “settle down” and the system
is then sampled for a certain finite length of time Ts (we use frequency-time language only for simplicity here, the same
ideas apply to wavelength-spatial domain problems).

11.3 Syntax of qfft.mac Functions

FAST FOURIER TRANSFORM UTILITIES
nyquist (ns,fs)
sample (expr,var,ns,dvar)
vf (flist,dvar)
kg (glist)
fchop (expr)
fchop1 (expr,small)
current_small ()
setsmall (val)
spectrum ( glist, nlw, ymax, k1,k2 )

1. nyquist (ns, fs), given ns, the number of signal
samples, and fs, the sampling frequency,
returns the list [dt,knyq,fnyq,df] where dt is the time
interval between function samples (dt = 1/fs), knyq in the Nyquist
integer (knyq = ns/2), fnyq is the Nyquist frequency
(fnyq = fs/2 = knyq*df), and df is the frequency
resolution (df = fs/ns) for the output of fft.
For given ns, the values of df and dt are linked
by the equation df*dt = 1/ns.

Thus nyquist(8,8); returns the list
[0.125, 4, 4.0, 1.0], and also prints out:

sampling interval dt = 0.125
Nyquist integer knyq = 4
Nyquist freq fnyq = 4.0
freq resolution df = 1.0

2. sample (expr, var, ns, dvar) constructs a list of ns
floating point samples F(m*dvar),
[F(0), F(dvar), F(2*dvar),...,F((ns-1)*dvar) ],
given the expression expr depending on var.

sample(cos(t),t,16,1) returns a list of 16 values
of cos(t) at intervals dt = 1, with list element
number 1 holding cos(0), element number 2 holding
cos(dt) = cos(1), etc.

A signal sample list to be used with fft(flist) should
have a length ns which is 2 raised to some integer power,
2ˆ3 = 8, 2ˆ4 = 16,... If your experimental signal sample
size does not have such a power of 2 length, you should pad
the sample list with extra zeros.

3. vf ( flist, dvar ), given a list of function samples flist,
consisting of ns values, and the step size dvar, returns a

list of the form (if dvar = dt, say),
[ [0, F(0)],[dt, F(dt)],...,[(ns-1)*dt, F( (ns-1)*dt)]]
useful for a plot.
For example, if the length of flist is 8, vf (flist, 1) returns the list

[ [0, F(0)], [1, F(1)], ... , [7, F(7)] ]

4. kg ( glist ) constructs a list of [ k, abs( g(k*df) ) ]
for k = 0, 1, ..., knyq, which can be used with plot2d.

knyq is the Nyquist integer, knyq = ns/2,
where ns is the number of function samples and also the
length of flist and glist. fchop(abs(glist[j])) is
used to be able to plot real numbers and set tiny

numbers to zero.

glist is the fast Fourier transform list of complex amplitudes
produced by glist : fft( flist ).

Using inverse_fft( glist ) should produce flist again to within
floating point errors. Since floating point errors will also
introduce tiny imaginary numbers in inverse_fft ( fft (flist) )
(if flist is real), you can use realpart(...) to recover
a list of real numbers.

5. fchop(expr) or fchop(list) sets tiny floating point
numbers (with package default, less than 10ˆ(-13) ) to zero.

6. fchop1(expr,small) is used to override the default value of
the small chop value with your desired value.

Example: fchop1(s1, 1.0e-3) to set numbers smaller than
10ˆ(-3) to zero in the expression or list s1.

7. current_small() returns the current default small chop value.

8. setsmall(val) allows you to set a new value for the
current default small chop value; use floating point
numbers like 2.0e-3 or 2.0E-3.

9. spectrum (glist, nlw, ymax ) creates a histogram of the frequency
spectrum implied by glist = fft ( flist ), with line width nlw

and vertical canvas height ymax. The range of integers
k is 0 <= k <= knyq. The frequency associated with a given
line is given by f = k*df, where df is the frequency
resolution df = fs/ns. ns is the total number of signal samples
and fs is the sampling frequency: fs = 1/dt (dt is the time
interval between signal samples).

spectrum (glist, nlw, ymax, k1, k2) restricts the plot to
the range k1 <= k <= k2.

11.4 The Discrete Fourier Transform Derived via a Numerical Integral Approximation

Review of Trapezoidal Rule

If we let f0 = f(a), f1 = f(a + h), f2 = f(a + 2h), ... and fN = f(b) = f(a + Nh), then the trapezoidal rule approx-
imation is ∫ b

a
f(x)dx ≈ h

2
(f0 + 2 f1 + 2 f2 + · · ·+ 2 fN−1 + fN) (11.27)

where b = a + Nh defines h = (b− a)/N. If we now specialize to functions such that f(a) = f(b) then the trapezoidal
rule reduces to

∫ b

a
f(x)dx ≈ h (f(a) + f(a + h) + f(a + 2h) + · · ·+ f(b− 2h) + f(b− h)) (11.28)

with h = (b− a)/N.

If we now make the replacements x → t, a → 0, b → T, h = (b− a)/N → T/N = ∆t,
(b− h) → T− (T/N) = (N− 1)∆t, then

∫ T

0
f(t)dt ≈ (T/N) (f(0) + f(∆t) + f(2∆t) + · · ·+ f( (N− 1)∆t) ) (11.29)

or ∫ T

0
f(t)dt ≈ ∆t

N−1∑

m=0

f(m∆t) (11.30)

where ∆t = T/N and assuming f(0) = f(T).

A Path to the Discrete Fourier Transform

If we knew the value of a signal F(t) at all moments of the interval 0 ≤ t ≤ T then we could evaluate the Fourier
coefficients

Ck =
1
T

∫ T

0
F(t) e−2π i k t/T dt (11.31)

(k is an integer) in terms of which the signal could be represented as the sum

F(t) =
∞∑

k=−∞
Ck e2π i k t/T (11.32)

which would be a complex form of Fourier series expansion in terms of an infinite number of Fourier coefficients Ck. We
have adopted here sign and prefactor conventions which will lead us to Maxima’s fast Fourier conventions.

Now assume we only know the signal at N discrete values F(m∆t), where m = 0,1,2, ...,N− 1, and that F(t) = f(t + T).
We can then approximate the integral in Eq.(11.31) using the trapezoidal approximation expressed by Eq.(11.30).

Ck ≈ 1
T

(T/N)
N−1∑

m=0

F(m∆t) e−2π i km∆t/T (11.33)

Defining a “frequency resolution” δf given by

δf =
1
T

=
1

∆tN
=

fs
N

(11.34)

in which fs = 1/∆t is the “sampling frequency”, and making the replacements T → 1/δf , Ck → G(k δf)

G(k δf) =
1
N

N−1∑

m=0

F(m∆t) e−2π i (k δf) (m∆t) (11.35)

With fk = k δf and tm = m∆t, we can write this as

G(fk) =
1
N

N−1∑

m=0

F(tm) e−2π i fk tm (11.36)

which we recognise as the same as Eq.(11.15). With only N values of F(tm) we can only determine N values of G(fk),
which by convention we take to be for the values of the integer k = 0,1,2, ...,N− 1.

We see from Eq.(11.34) that

δf ∆t =
1
N

(11.37)

which allows Eq.(11.35) to be written as

G(k δf) =
1
N

N−1∑

m=0

F(m∆t) e−2π i km/N (11.38)

which reproduces our starting point Eq.(11.12) in Sec. 11.2.

The Inverse Discrete Fourier Transform

Although the discrete Fourier transform, Eq.(11.38), is an approximation which gets better as N increases for fixed T (or since
Eq.(11.34) indicates that the latter condition is equivalent to the ratio fs/N being fixed, gets better as N and fs are each increased in
the same ratio), the inversion formula involves no further approximations.

Multiplying both sides of Eq.(11.38) by exp(2π i k n/N) and then summing both sides over k, and then interchanging the order of
the k and m summations on the right hand side, results in

N−1∑

k=0

G(k δf) e2 π i k n/N =
1
N

N−1∑
m=0

F(m∆t)
N−1∑

k=0

e2 π i k (n−m)/N =
1
N

N−1∑
m=0

F(m∆t) (N δm,n) = F(n∆t) (11.39)

which reproduces our discrete inverse Fourier transform formula Eq.(11.13), since n is an arbitrary integer.

That the sum over k is equal to zero if m 6= n can be seen by letting l = n−m and using eka = (ea)k and recognising that we
have a simple geometric sum. We can also let Maxima confirm this as follows:

(%i1) declare([k,l,N],integer)$
(%i2) sum (exp(2*%pi*%i*k*l/N),k,0,N-1),simpsum;

2 %i %pi l
%e - 1

(%o2) ----------------
2 %i %pi l
----------

N
%e - 1

(%i3) %,demoivre;
(%o3) 0

In the last step the exponentials of complex arguments are converted into their trig function equivalents using

ei θ = cos θ + i sin θ (11.40)

11.5 Fast Fourier Transform References
We have consulted the treatment of the fast Fourier transform in the following books:

1. Mathematica for the Sciences, Richard E. Crandall, Addison-Wesley, 1991

2. A First Course in Computational Physics, Paul L. Davies, John Wiley, 1994

3. Applied Mathematica: Getting Started, Getting it Done, William T. Shaw and Jason Tigg, Addison-Wesley, 1994


